Business Negotiation Definition Language

Rustam Tagiew
Alumni of TU Bergakademie Freiberg and Uni Bielefeld
rustam@tagiew.de, www.tagiew.de

Abstract. The target of this paper is to present an industry-ready prototype software for general game playing. This software can also be used as the central element for experimental economics research, interfacing of game-theoretic libraries, AI-driven software testing, algorithmic trade, human behavior mining and simulation of (strategic) interactions. The software is based on a domain-specific language for electronic business to business negotiations – SIDL3.0. The paper also contains many examples to prove the power of this language.

Keywords: General Game Playing, Experimental Economics, Game Theory, Price Negotiations, AI driven Software Testing, Algorithmic Trade, Mechanism Design, Electronic Negotiations, B2B, behavior mining, Domain-Specific Languages

1 Motivation

This paper describes the scientific background of a ready implemented prototype software available under github.com/Yepkio/sidl [1].

The issue discussed in this paper provides improvements to many fields simultaneously and facilitates their interconnections. Since the extent of this issue is so over-sized to almost any reader non-proficient or only partially proficient in the overall carpet of research fields, the paper proceeds along the leitmotif of a business application. The concrete business applications are the electronic business to business (B2B) negotiations as Friedrich Michael described in his paper [2].

These electronic B2B negotiations (EB2BN) have equilibria in game-theoretic terms, which should be observed in case of rationality and perfect reasoning. Although this analytical solution by game-theoretic math is rarely the case in real world, certain convergence towards this solution exists. Game-theoretic solution also provides a rough estimate of the soundness for a certain EB2BN. For instance, when a buying company sets up a complicated EB2BN with multiple sellers, it traces certain goals of reducing the price or just of receiving and spreading information. The software is desired to be able to make an automatic game-theoretic analysis of an EB2BN defined in a domain-specific language. The seminal work solving this kind of problems is Gala [3].

EB2BNs are implemented by computer systems. General Game Playing (GGP) means development of AI systems, which are based on a domain-specific language for games and therefore can deals with games in general terms [4]. For EB2BNs, this means development of a business negotiation engine, which runs on a domain-specific language for EB2BNs. In GGP, it is called game management and calculates the game
A very interesting topic for EB2BN is algorithmic trade. The selling companies hereby employ automated negotiators. The automated negotiators aka agents receive the definition of EB2BN like in GGP in advance. Then, the agents can make intelligent decisions in favor of their shepherd. Agents will not be the case in the beginning. Human employees of the sellers will be put in front of a user interface to the EB2BN. Those humans have certain typically human behavior patterns, which are studies using the data from the laboratory experiments. And those experiments are carried out by experimental economists. A domain-specific language would greatly facilitate transfers from field to laboratory and back. Human behavior mining would greatly profit from clearly defined EB2BN setups. Current industry solutions do not use any GGP style engines for EB2BNs. Current EB2BN systems are hard-coded. Programming of systematic testing of these software pieces requires deep knowledge of the concrete EB2BNs. Testing software, which is based on a domain-specific language would be much easier to handle.

And finally, offline simulation of EB2BNs is hardly imaginable without a domain-specific language. The goal of an offline simulation is to test a concrete EB2BN with humans or/and agents. If this EB2BN does return the needed results, the simulation should be easily rerun with an adjusted EB2BN.

Section 2 briefly summarizes prehistory of this development. Section 3 is hopefully the easiest way to present the formal background of SIDL3.0 – a language usable for EB2BN. Section 4 presents the state transition calculating algorithm for SIDL3.0. Examples of game definitions in SIDL3.0 are given in section 5.

2 Related Work

A good and still prevailing review [5] presents languages for game-theoretic problems. Languages and frameworks for setting up experiments in behavioral economics also are numerous [6].

Strategic Interaction Definition Language (SIDL) has a decade long prehistory. Strategic interaction is a more clear term for a game as meant in Game Theory. SIDL was first introduced in 2009 [7] as a universal language for games of imperfect information. Then, it was significantly improved as SIDL2.0 [8, p.98–105]. This paper presents SIDL3.0, which is minor improvement of SIDL2.0 adding unlimited spaces for actions.

3 Language Definition

SIDL [ˈzaɪdɔl] is a first-order logic definition of a game. SIDL structure is derived from STRIPS formalism [9] and situation calculus [10]. The formal definition of SIDL3.0 is given in Def[1]. Like in the language for planning tasks PDDL [11], SIDL describes the state of the world by a set of facts. From SIDL2.0 on, these facts are lists of literals,
called words. Closed world assumption applies for SIDL. In chess for instance, following facts mean that white knight is on c4, black queen is on e6 and nothing else is on the board:

\[[\text{white, knight, c, 4}]\]
\[[\text{black, queen, e, 6}]\]

According to chess rules, there are multiple squares on this board, where the knight and the queen can move in next turn from this **initial state** (Def.1, 2). Every of such moves is an **action** in SIDL (Def.1, 3). Together with a definition of a state transition (Def.1, 4) to every of these actions or their combinations, it is already a definition of a planning task, if 'players' and 'switches' are blanked out.

**Definition 1 (SIDL3.0).** A game description in SIDL3.0 consists of following elements, whereby $\Sigma$ is a finite set. $\Sigma$ consists of a finite set of symbols and a finite subset of real numbers:

1. $S \subset \{w \circ m : w \subset \Sigma^* \land m \in \mathbb{R}^{|N|}\}$ a set of states. Every state is set of words consisting of symbols $\Sigma$ and an assignments for players' accounts.
2. $s_{\text{go}} \in S$ initial state.
3. $A \subset \Sigma^*$ a set of actions, every action is a word consisting of symbols from $\Sigma$.
4. $\triangledown : S \times A^m \rightarrow S$ Definition of state transitions for a SIDL-automaton. $m$ is the number of simultaneously applicable actions, which is bigger than 0 and does not exceed the number of legal switches.
5. $N \subset \Sigma^*$ a set of players, which are words consisting of symbols from $\Sigma$.
6. $I \subset \Sigma^*$ a set of switches.
7. $L : I \times \Sigma^{**} \rightarrow \mathbb{N}^*$ a truth function to define the legality of a switch. Terminal states have an empty set of legal switches.
8. $P : I \times \Sigma^{**} \rightarrow A^*$ a function to define a set of actions for every switch depending on state. This function is bijective.
9. $O : \{i \circ w \mapsto n : w \subset \Sigma^* \land i \in I \land n \in (N \cup (\mathbb{R}^{|P(l,w)|}))\}$ an attribution of a player or of a distribution to a switch depending on state.
10. $H : \Sigma^* \times N^*$ a truth function for hidden words.
11. Following applies:

\[\forall n \in N : \forall s, s' \in S : \forall l \in I :\]

\[\{\{w \in s : \neg H(w, n)\} = \{w \in s' : \neg H(w, n)\}\} \land (n = O(l, s)) \Rightarrow n = O(l, s') \land (P(l, s) = P(l, s')) \land (L(l, s) \Leftrightarrow L(l, s'))\]

The elements 5-11 of Def.[1] are needed, since we have more than one 'planner' in a game. Element 5 defines players as words similar to facts – it simplifies the definition of general rules.

Elements 6-8 define the sets of mutually exclusive actions as *switches*. A switch is basically a set of actions, from where only one can be applied in a state transition. If game rules require multiple simultaneous actions, then actions from multiple switches should be used. There is also a function, which defines *legality* of a switch depending on the game state. The set of actions belonging to a switch also depends on the game
state – it adds casual flexibility to SIDL3.0.

Element $O$ defines either an ownership of a switch by player or an assignment to a probability distribution over the available actions. This probability distribution is needed for games, which include random events. Like the set of actions itself, the distribution also depends on the game state.

The game-theoretic notion of imperfect information should not be confused with incomplete information. Not seeing parts of the game state is called imperfect, and not seeing parts of the game rules is called incomplete. $H$ defines imperfect information for SIDL by hiding facts from players. This is a delicate matter in SIDL – the hidden facts should neither be revealed by legality of switches, nor by the sets of actions from switches for a certain player (Def 11). Otherwise a player might deduce the hidden facts – they are not hidden anymore.

Finally, to predict rational or rationally motivated behavior of players given the game rules, we need some kind of preferences being defined in the game rules. This is done by ‘special facts’ – the players’ accounts. Every change of the game state might change them by payoffs. A rational player is expected to maximize his/her account. It can also be a case that the real payoffs are not known.

The syntax of SIDL is based on ISO Prolog [12]. ISO Prolog is a logic programming language – it has in its pure form neither loops nor conditionals. There are rules consisting of a rule head and of a rule body. Rule body is a condition for the rule head. Following language elements are available for game description in SIDL3.0:

- All ISO Prolog operators for lists’ and numbers’ manipulations are allowed. Operators game, init, hidden, legal, switch, unlimited, owned, default, do and payoff are keywords of SIDL3.0 and only allowed in rule heads. player, fact, create, delete, tocreate, todelete and does also are keywords and are only allowed in rule bodies.
- game(NAME) – NAME is the name of the game.
- init(W):-Condition – such rules define the words of the initial state. Every word $W = \{a, \ldots\}$ is represented as a list.
- player(N) – this operator returns a valid player $N$.
- fact(W) – this returns valid word $W$.
- hidden(W, P):-Condition – word $W$ is hidden for player $P$.
- Condition may contain atoms with operator player.
- legal(I):-Condition – list $I$ is legal switch in the actual state. Condition may contain atoms with operator player and fact.
- owned(I, D):-Condition – switch $I$ is attributed to a player or to a distribution $D$. A distribution is either a list of probabilities or an expression $equal(X)$, which is a uniform distribution over $X$ actions. Condition may contain atoms with operators player and fact.
- switch(I, A):-Condition – list $A$ is one of possible actions for switch $I$. Condition may contain atoms with operators player and fact.
- unlimited(I, T):-Condition – if a list of actions for a switch $I$ is unreasonable to be created due to its size, $T$ can be defined as its template structure by this operator. Condition may contain atoms with operators player and fact.
- create(W) – word $W$ will be added in next turn.
- delete(W) – word $W$ will be deleted in next turn.
Business Negotiation Definition Language

- `tocreate(W)` – returns a word W, which will be added in next turn.
- `todelete(W)` – returns a word W, which will be deleted in text turn.
- `does(I, A)` – returns the action A for the switch I.
- `do(A):-Condition` – definition of the action A. Condition may contain atoms with operators `player`, `fact`, `create`, `delete` and `does`.
- `default(I, A):-Condition` – default action A of the switch I. Condition may contain operators `player` and `fact`.
- `init(N, M):-Condition` – player accounts in the initial state. List N is a player and M is its account balance.
- `payoff(N, R):-Condition` – payoff R will be added to the account of player N. Condition may contain atoms with operators `player`, `fact`, `tocreate`, `todelete` and `does`.

4 Chronons versus Interrupts

Based on a preliminary decision of allowing or denying simultaneous actions by multiple players, there are two ways of game management based on SIDL. The first is based on chronons as time quanta between state transitions, during which all players’ actions are considered simultaneous. The second is an interrupt based approach with every valid action triggering a state transition and no simultaneous actions. For the interrupt based approach, m from element 4 of Def[1] equals always 1. The chronon based approach offers obviously more advantages – the interrupt based approach is skipped from consideration in this paper.

Alg[1.1] shows the game management algorithm for the chronon based approach. `lcall` stands for calls into the Prolog engine. `assert` and `retract` add and delete facts from the game state. The algorithm runs in a loop as long as there are legal switches. Every loop run lasts one chronon.

There are cases in business application, where the rules have to be changed during a game run [2]. For instance, new players have to be added. This extraordinary situation equals to a restart of a complete game, reassignment of agents into the new player roles and sending them the new game rules.

5 Examples

Following examples will present variants of definitions for commonly known games. Some of less important details of the game rules will not be commented. The reader is welcome to dive into the code and maybe find better definitions for those games.

In game Nim, players have to subtract in each turn between 1 and 3 items from a pool. Players rotate and the one, who is last at subtracting loses.

```
nim.
init([alice, 10]).
init([alice], 0.0).
init([bob], 0.0).
```
Algorithm 1.1: Game management based on SIDL3.0 [8, p.106].

```plaintext
1 foreach F in lcall(init(F)).solutions do
2     lcall(assert(fact(F)));
3 end

4 foreach (N, M) in lcall(init(N, M)).solutions do
5     lcall(assert(account(N, M)));
6 end

7 while lcall(legal(_)) do
8     while chronon_not_expired do
9         assert_does_for_a_valid_incomming_command;
10    end
11    foreach (I,D) in lcall(legal(I)∧owned(I,D)∧dist(D)).solutions do
12        assert_randomly_does(I,D);
13    end
14    foreach (I,A) in lcall(legal(I)∧(does(I,A)∨default(I,A))).solutions do
15        lcall(do(A));
16    end
17    foreach (N,P) in lcall(goal(N,P)).solutions do
18        lcall(retract(account(N,M))∧assert(account(N,M+P)));
19    end
20    foreach (F) in lcall(tocreate(F)∨todelete(F)).solutions do
21        send_unhiden_changes_to_players(F);
22    end
23    foreach (N,M) in lcall(account(N,M)).solutions do
24        send_to_players(N,M);
25    end
26    foreach (F) in lcall(todelete(F)).solutions do
27        lcall(retract(todelete(F))∧retract(fact(F)));
28    end
29    foreach (F) in lcall(tocreate(F)).solutions do
30        lcall(retract(tocreate(F))∧assert(fact(F)));
31    end
32    lcall(retractall(does(_,_)));
33 end
34 end_the_game;
```
Muddy Children Puzzle (MCP) is a frequently cited logic puzzle. In this definition, there are 5 children and $n$ of them get muddy faces by chance. Every child sees other children’s faces, but he/she does not see the own one. Later, children stand in a row—they are given an infinite sequence of opportunities to step forward in order to signal own muddy face. If own face is muddy, every missed opportunity for stepping forward is bad ($-1$). If own face is not muddy, stepping forward is much worse ($-10000$). All $n$ muddy children step forward at $n$-th opportunity in case of rational and perfectly reasoning children.

name(mcp).
children([alice, bob, charly, david, eric]).
makedirty([]).
makedirty([C | Cs]) :- create([dirty, C]), makedirty(Cs).
init([start]).
init([C], 0.0) :- children(Cs), member(C, Cs).
hidden([dirty, C], [C]) :- player([C]).
legal([dirt]) :- fact([start]).
legal([C]) :-
    player([C]), not(fact([start])), not(fact([_, stepped])).
switch([dirt], [dirt | SCs]) :-
    children(Cs), getsubset(SCs, Cs), not(SCs = []).
switch([C], [C, S]) :-
    player([C]), member(S, [stay, step]).
owned([dirt], equal(N)) :-
    children(Cs), length(Cs, L), N is (2^L)-1.
owned([C], [C]) :- player([C]).
default([C], [stay]) :- player([C]).
do([dirt | DCs]) :-
    fact([start]), delete([start]), makedirty(DCs).
do([_, stay]).
do([C, step]) :- not(fact([start])), create([C, stepped]).
payoff([C], -1) :-
    not(fact([start])), not(tocreate([C, stepped])).
payoff([C], 100.0):-

This is the well-known Rock Paper Scissors in SIDL3.0. There are 10 rounds in this game definition. Every round lasts 3 chronons. Already made actions from the previous round are shown to everybody. The default action for both players is Rock.

name(rps).
gestures([rock, paper, scissors]).
beats(paper, rock).
beats(rock, scissors).
beats(scissors, paper).
opponent(role1, role2).
opponent(role2, role1).
giventime(3).
givenrounds(10).
gesture(P, G):-
    not(does([P], _)), fact([chosen, P, G]).
gesture(P, G):-
    does([P], [P, wait]), fact([chosen, P, G]).
gesture(P, G):-
    does([P], [P, G]).
regesture(P, G):-
    gesture(P, G), not(G = O),
    delete([chosen, P, O]), create([chosen, P, G]).
madegestures:-
    gesture(role1, G1), create([made, role1, G1]),
    gesture(role2, G2), create([made, role2, G2]).
init([gameon]).
init([chosen, P, rock]):- opponent(P, _).
init([rounds, 10]):- givenrounds(10).
init([timer, X]):- giventime(X).
init([P], 0.0):- opponent(P, _).
hidden([chosen, P, _], [O]):- opponent(P, O).
legal([P]):- fact([gameon]), player([P]).
legal([timer]):- fact([timer, X]), X > 0.
legal([round]):- fact([timer, 0]), fact([gameon]).
switch([P], [P, wait]):- player([P]).
switch([P], [P, G]):- player([P]), gestures(Gs), member(G, Gs).
switch([timer], [timer]).
switch([round], [round]).
owned([P], [P]):- player([P]).
owned([timer], equal(1)).
owned([round], equal(1)).
do([_, wait]).

do([P, G]):-
    player([P]), fact([chosen, P, O]),
    not(O = G),
    delete([chosen, P, O]), create([chosen, P, G]).

do([timer]):-
    fact([timer, X]), X > 0, XMM is X - 1,
    delete([timer, X]), create([timer, XMM]).

do([round]):-
    fact([rounds, X]),
    X > 1,
    delete([timer, 0]), delete([rounds, X]),
    NX is X - 1,
    create([rounds, NX]),
    givetime(T), create([timer, T]),
    madegestures.

do([round]):-
    fact([rounds, 1]), madegestures, delete([gameon]).

payoff([P], 1.0):-
    does([round], [round], gesture(P, G1),
    opponent(P, O), gesture(O, G2), beats(G1, G2).

payoff([P], 0.0):- opponent(P, _).

This example EB2BN has a starting price and players have to overbid it. This definition adds all bids to game state and calculates the highest of them for every new state. Operator unlimited is used to define bids as real numbers.

bidders([alice, bob, clara, david]).
namename(priceNegotiation).
init([startprice, 10.0]).
init([A], 0.0):- bidders(Rs), member(A, Rs).
legal(R):- player(R).
unlimited(R, [wait]):- player(R).
unlimited([A], [A, (price,double)]):- player([A]).
leadingprice(B):-
    not(fact([bid,_,_])), fact([startprice,B]).
leadingprice(B):-
    fact([bid,_,B]), findall(P, fact([bid,_,P]),Ps),
    maxmember(Ps,B).
owned(R, R):- player(R).
switch([A], [A, T]):- player([A]), leadingprice(B), T > B.
switch(R, [wait]):- player(R).
default(R, [wait]):- player(R).
do([wait]).
do([A, P]):- create([bid, A, P]).

payoff(R, 0.0):- player(R).
Chess has four types of actions – move, take, castle and promote. Castling can be left or right. The precondition for castling can be irreversibly invalidated for both players. Therefore in addition to the positions of men, chess needs facts representing the castling precondition. This definition of chess does not contain a definition of a draw nor a limit on turns.

name(chess).
opposite(white, black).
opposite(black, white).
xaxis([a,b,c,d,e,f,g,h]).
direction(white, 1).
direction(black, -1).
forpromotion([queen, knight, rook, bishop]).
directions(rook, [[1, 0], [0, 1], [-1, 0], [0, -1]]).
directions(bishop, [[1, 1], [1, -1], [-1, 1], [-1, -1]]).
directions(queen, A):-
    directions(rook, R), directions(bishop, B), append(R, B, A).
position(knight, X, Y):- member(X, [-2, 2]), member(Y, [-1, 1]).
position(knight, X, Y):- member(Y, [-2, 2]), member(X, [-1, 1]).
position(king, X, Y):- directions(queen, A), member([X, Y], A).
xvalue(Pre, Next, Inc):-
    xaxis(Xs), nth1(I, Xs, Pre), II is I + Inc, nth1(II, Xs, Next).
yvalue(Pre, Next, Inc):- Next is Pre + Inc, yvalue(Next).
row(white, 1, 1).
row(black, 8, 1).
row(black, 7, 2).
man(rook, [a, h], 1).
man(knight, [b, g], 1).
man(bishop, [c, f], 1).
man(queen, [d], 1).
man(king, [e], 1).
man(pawn, Xs, 2):- xaxis(Xs).
promote(Color, Y, Man):-
    opposite(Color, Opp), row(Opp, Y, 1),
    forpromotion(HMs), member(Man, HMs).
init([Color, Man, Y]):-
    row(Color, Y, Row), man(Man, Xs, Row), member(X, Xs).
init([white]).
init([white, fortifiable]).
init([black, fortifiable]).
init([white], 0.0).
init([black], 0.0).
legal([C]):- fact([C, king, _, _]), player([C]), fact([C]).
contrline(_, X, _, [DX, _], []):- not(xvalue(X, _, DX)).
contrline(_, _, Y, [_, DY], []) :- not(yvalue(Y, _, DY)).
contrline(Color, X, Y, [DX, DY], []) :-
xvalue(X, X1, DX), yvalue(Y, Y1, DY), fact([Color, _, X1, Y1]).
contrline(Color, X, Y, [DX, DY], [[X1, Y1]]) :-
xvalue(X, X1, DX), yvalue(Y, Y1, DY),
opposite(Color, Opp), fact([Opp, _, X1, Y1]).
contrline(Color, X, Y, [DX, DY], [[X1, Y1] | Line]) :-
xvalue(X, X1, DX), yvalue(Y, Y1, DY),
not(fact([_, _, X1, Y1])),
contrline(Color, X1, Y1, [DX, DY], Line).
controlled(Color, pawn, X, Y, X1, Y1) :-
  fact([Color, pawn, X, Y]),
direction(Color, I),
yvalue(Y, Y1, I), member(D, [1, -1]), xvalue(X, X1, D).
controlled(Color, Man, X, Y, X1, Y1) :-
directions(Man, Ds),
fact([Color, Man, X, Y]),
member(D, Ds),
contrline(Color, X, Y, D, NP),
member([X1, Y1], NP).
controlled(Color, Man, X, Y, X1, Y1) :-
  position(Man, DX, DY),
fact([Color, Man, X, Y]),
xvalue(X, X1, DX), yvalue(Y, Y1, DY),
not(fact([Color, _, X1, Y1])).
switch([Color], [Color, pawn, X, Y, X2]) :-
  fact([Color, pawn, X, Y]), row(Color, Y, 2),
direction(Color, I),
yvalue(Y, Y2, I * 2), yvalue(Y, Y1, I),
not(fact([_, _, X, Y1])),
not(fact([_, _, X, Y2])).
switch([Color], [Color, pawn, X, Y, Y1]) :-
  fact([Color, pawn, X, Y]),
direction(Color, I), yvalue(Y, Y1, I),
not(promote(Color, Y1, _)),
not(fact([_, _, X, Y1])).
switch([Color], [Color, pawn, X, Y, X1, Man]) :-
  fact([Color, pawn, X, Y]),
direction(Color, I), yvalue(Y, Y1, I),
promote(Color, Y1, Man),
not(fact([_, _, X, Y1])).
switch([Color], [Color, pawn, X, Y, X1, Y1, Man]) :-
  controlled(Color, pawn, X, Y, X1, Y1),
opposite(Color, Opp),
fact([Opp, _, X1, Y1]),
promote(Color, Y1, Man)).
switch([Color], [Color, pawn, X, Y, X1, Y1]):-
  controlled(Color, pawn, X, Y, X1, Y1),
  opposite(Color, Opp), fact([Opp, _, X1, Y1]),
  not(promote(Color, Y1, _)).

switch([Color], [Color, Man, X, Y, X1, Y1]):-
  controlled(Color, Man, X, Y, X1, Y1),
  not(Man = pawn), not(Man = king).

switch([Color], [Color, king, X, Y, X1, Y1]):-
  controlled(Color, king, X, Y, X1, Y1),
  opposite(Color, Opp),
  not(controlled(Opp, _, _, _, X1, Y1)).

switch([Color], [Color, castle, right, Y]):-
  fact([Color, fortifiable]),
  fact([Color, rook, h, 1]),
  row(Color, Y, 1), not(fact([_, _, f, Y])),
  not(fact([_, _, g, Y])), opposite(Color, Opp),
  not(controlled(Opp, _, _, _, f, Y)),
  not(controlled(Opp, _, _, _, g, Y)).

switch([Color], [Color, castle, left, Y]):-
  fact([Color, fortifiable]),
  fact([Color, rook, a, 1]), row(Color, Y, 1),
  not(fact([_, _, b, Y])), not(fact([_, _, c, Y])),
  not(fact([_, _, d, Y])), opposite(Color, Opp),
  not(controlled(Opp, _, _, _, b, Y)),
  not(controlled(Opp, _, _, _, d, Y)).

owned(C, C):- player(C).

changeturn(Color):-
  opposite(Color, Opp), delete([Color]), create([Opp]).
nocastle(Color, Man, X, Y):-
  opposite(Color, Opp), fact([Opp, king, XK, YK]),
  controlled(Color, Man, X, Y, XK, YK), delete([Opp, fortifiable]).
nocastle(Color, _, X, Y):- opposite(Color, Opp),
  fact([Opp, king, X, Y]), delete([Color, fortifiable]).
nocastle(_, _, _):- delete([Color, fortifiable]).

do([Color, Man, X, Y, NX, NY]):-
  fact([C, M, NX, NY]),
  changeturn(Color), nocastle(Color, Man, NX, NY),
  delete([C, M, NX, NY]), delete([Color, Man, X, Y]),
  create([Color, Man, NX, NY]).

do([Color, Man, X, Y, NX, NY]):-
  changeturn(Color),
  delete([Color, Man, X, Y]), nocastle(Color, Man, NX, NY),
  create([Color, Man, NX, NY]).

do([Color, Man, X, Y, NX, NY, New]):-
  changeturn(Color),
  delete([Color, Man, X, Y]), nocastle(Color, Man, NX, NY),
  create([Color, New, NX, NY]).
do([Color, castle, right, Y]) :- changeturn(Color),
    delete([Color, king, e, Y]), delete([Color, rook, h, Y]),
    delete([Color, fortifiable]),
    create([Color, king, g, Y]), create([Color, rook, f, Y]).

do([Color, castle, left, Y]) :- changeturn(Color),
    delete([Color, king, e, Y]), delete([Color, rook, a, Y]),
    delete([Color, fortifiable]),
    create([Color, king, d, Y]), create([Color, rook, c, Y]).

payoff([Color], 1.0):-
    opposite(Color, Opp), todelete([Opp, king, _, _]).

payoff([Color], -1.0):- todelete([Color, king, _, _]).

References

1. Yepkio: Strategic interaction description language. [https://github.com/Yepkio/sidl](https://github.com/Yepkio/sidl) (2019)
2. Michael, F., Ignatov, D.I.: General game playing b-to-b price negotiations. In: Proceedings of the Fifth Workshop on Experimental Economics and Machine Learning (EEML 2019) co-located with the 7th International Conference on Applied Research in Economics (iCare7 2019). Volume 2479 of CEUR Workshop Proceedings., Perm, Russia, CEUR-WS.org (2019)
3. Koller, D., Pfeffer, A.: Representations and solutions for game-theoretic problems. Artificial Intelligence 94 (1997) 167–215
4. Genesereth, M.R., Love, N., Pell, B.: General game playing: Overview of the aaai competition. AI Magazine 26(2) (2005) 62–72
5. Tagiew, R.: Beyond analytical modeling, gathering data to predict real agents’ strategic interaction. SKAD’11 – Soft Computing Applications and Knowledge Discovery (2011) 113
6. Balietti, S.: nodegame: Real-time, synchronous, online experiments in the browser. Behavior Research Methods 49(5) (Oct 2017) 1696–1715
7. Tagiew, R.: General game management agent. CoRR abs/0903.0353 (2009)
8. Tagiew, R.: Strategische Interaktion realer Agenten: Ganzheitliche Konzeptualisierung und Softwarekomponenten einer interdisziplinären Forschungsinfrastruktur. PhD thesis, TU Bergakademie Freiberg (2011)
9. Fikes, R., Nilsson, N.: Strips: a new approach to the application of theorem proving to problem solving. Artificial Intelligence 2 (1971) 189–208
10. McCarthy, J., Hayes, P.J.: Some philosophical problems from the standpoint of artificial intelligence. Machine Intelligence 4 (1969) 463–502
11. McDermott, D.V.: The 1998 ai planning systems competition. AI Magazine 21 (2000) 35–55
12. Deransart, P., Ed-Dbali, A., Cervoni, L.: Prolog: The Standard. Springer-Verlag (1996)